
FORMATTED LISTING
; ####################################################################################################
; #   PROJECT  :  Apple /// SOS Serial Printer Driver 1.30 (6502 Assembly Source Code)
; #   FILE NAME:  SERPRINT.text
; ####################################################################################################

000001                  .TITLE       "SOS Serial Printer Driver"
000002                  .NOPATCHLIST
000003                  .NOMACROLIST
000004
000005  ;-----------------------------------------------------------------------
000006  ;
000007  ;               SOS Serial Printer Driver
000008  ;
000009  ;               Copyright (C) 1983 by Apple Computer Inc.
000010  ;               All Rights Reserved
000011  ;
000012  ;
000013  ;       Revisions:
000014  ;
000015  ;       1.00    14-Nov-80
000016  ;
000017  ;       1.10    14-Apr-81
000018  ;               Bug fixes:
000019  ;               Switch to 1 MHz for all ACIA references.
000020  ;               Check buffer count and delay count for write completion.
000021  ;
000022  ;       1.30    05-Jan-83
000023  ;               Bug fixes:
000024  ;               Add XMIT flag for improved communications between Driver
000025  ;                  and Interrupt Handler.
000026  ;
000027  ;-----------------------------------------------------------------------
000028
000029  DEVTYPE         .EQU         41
000030  SUBTYPE         .EQU         01
000031  APPLE           .EQU         0001
000032  RELEASE         .EQU         1300
000033                  .PAGE
000034  ;-----------------------------------------------------------------------
000035  ;
000036  ;  The macro SWITCH performs an N way branch based on a switch index.  The
000037  ;  maximum value of the switch index is 127 with bounds checking provided
000038  ;  as an option.  The macro uses the A and Y registers and alters the C,
000039  ;  Z, and N flags of the status register, but the X register is unchanged.
000040  ;
000041  ;               SWITCH  [index], [bounds], adrs_table, [*]
000042  ;
000043  ;       index   This is the variable that is to be used as the switch index.
000044  ;               If omitted, the value in the accumulator is used.
000045  ;
000046  ;      bounds   This is the maximum allowable value for index.  If index
000047  ;               exceeds this value, the carry bit will be set and execution
000048  ;               will continue following the macro.  If bounds is omitted,
000049  ;               no bounds checking will be performed.
000050  ;
000051  ;  adrs_table   This is a table of addresses (low byte first) used by the
000052  ;               switch.  The first entry corresponds to index zero.
000053  ;
000054  ;           *   If an asterisk is supplied as the fourth parameter, the
000055  ;               macro will push the switch address but will not exit to
000056  ;               it; execution will continue following the macro.  The
000057  ;               program may then load registers or set the status before
000058  ;               exiting to the switch address.
000059  ;
000060  ;-----------------------------------------------------------------------
000061  ;
000062                  .MACRO       SWITCH
000063                  .IF          "%1" <> ""                ;If PARM1 is present,
000064                  LDA          %1                        ;  Load A with switch index
000065                  .ENDC
000066                  .IF          "%2" <> ""                ;If PARM2 is present,
000067                  CMP          #%2+1                     ;  Perform bounds checking
000068                  BCS          $3579                     ;  on switch index
000069                  .ENDC
000070                  ASL          A
000071                  TAY
000072                  LDA          %3+1,Y                    ;Get switch address from table
000073                  PHA                                    ;  and push onto stack
000074                  LDA          %3,Y
000075                  PHA
000076                  .IF          "%4" <> "*"               ;If PARM4 is omitted,
000077                  RTS                                    ;  Exit to code
000078                  .ENDC                                  ;Otherwise, drop through
000079  $3579           .ENDM
000080                  .PROC        SERPRNT
000081                  .WORD        0FFFF
000082                  .WORD        66.
000083                  .ASCII       "Serial Printer Driver -- "
000084                  .ASCII       "Copyright (C) 1983 by Apple Computer Inc."

000085
000086  ;----------------------------------------------------------------------
000087  ;
000088  ;  Device Handler Identification Block
000089  ;
000090  ;----------------------------------------------------------------------
000091
000092  IDBLK           .WORD        0000                      ;Link to next device handler
000093                  .WORD        SP_MAIN                   ;Entry point address
000094                  .BYTE        8                         ;Length of device name
000095                  .ASCII       ".PRINTER       "
000096                  .BYTE        80,00,00                  ;Device, Slot & Unit numbers
000097                  .BYTE        DEVTYPE
000098                  .BYTE        SUBTYPE
000099                  .BYTE        00
000100                  .WORD        0000
000101                  .WORD        APPLE
000102                  .WORD        RELEASE
000103
000104
000105  ;----------------------------------------------------------------------
000106  ;
000107  ;  Device Handler Configuration Block
000108  ;
000109  ;----------------------------------------------------------------------
000110
000111                  .WORD        05                        ;Configuration block length
000112  DRATE           .BYTE        08                        ;Data Rate
000113  DFORMAT         .BYTE        22                        ;Data Format
000114  CRDELAY         .BYTE        00                        ;Carriage return delay
000115  LFDELAY         .BYTE        00                        ;Line feed delay
000116  FFDELAY         .BYTE        00                        ;Form feed delay
000117                  .PAGE
000118  ;----------------------------------------------------------------------
000119  ;
000120  ;  SOS Global Data & Subroutines
000121  ;
000122  ;----------------------------------------------------------------------
000123
000124  ALLOCSIR        .EQU         1913
000125  DEALCSIR        .EQU         1916
000126  SYSERR          .EQU         1928
000127
000128
000129  ;----------------------------------------------------------------------
000130  ;
000131  ;  SOS Error Codes
000132  ;
000133  ;----------------------------------------------------------------------
000134
000135  XREQCODE        .EQU         20                        ;Invalid request code
000136  XCTLCODE        .EQU         21                        ;Invalid control/status code
000137  XNOTOPEN        .EQU         23                        ;Device not open
000138  XNOTAVIL        .EQU         24                        ;Device not available
000139  XNORESRC        .EQU         25                        ;Resource not available
000140  XBADOP          .EQU         26                        ;Invalid operation for device
000141
000142
000143  ;----------------------------------------------------------------------
000144  ;
000145  ;  Hardware I/O Addresses
000146  ;
000147  ;----------------------------------------------------------------------
000148
000149  ACIADATA        .EQU         0C0F0                     ;ACIA data register
000150  ACIASTAT        .EQU         0C0F1                     ;ACIA status register
000151  ACIACMD         .EQU         0C0F2                     ;ACIA command register
000152  ACIACTL         .EQU         0C0F3                     ;ACIA control register
000153  E_REG           .EQU         0FFDF                     ;Environment register
000154  B_REG           .EQU         0FFEF                     ;Bank register
000155
000156
000157  ;----------------------------------------------------------------------
000158  ;
000159  ;  Miscellaneous Equates
000160  ;
000161  ;----------------------------------------------------------------------
000162
000163  TRUE            .EQU         80
000164  FALSE           .EQU         00
000165  ASC_LF          .EQU         0A
000166  ASC_FF          .EQU         0C
000167  ASC_CR          .EQU         0D
000168  BITON4          .EQU         10
000169  BITON7          .EQU         80
000170                  .PAGE
000171  ;-----------------------------------------------------------------------
000172  ;
000173  ;   SOS Device Handler Interface
000174  ;
000175  ;-----------------------------------------------------------------------
000176
000177  SOSINT          .EQU         0C0

000178  REQCODE         .EQU         SOSINT+0                  ;SOS request code
000179  BUFFER          .EQU         SOSINT+2                  ;Buffer pointer
000180  REQCNT          .EQU         SOSINT+4                  ;Requested count
000181  CTLSTAT         .EQU         SOSINT+2                  ;Control/status code
000182  CSLIST          .EQU         SOSINT+3                  ;Control/status list pointer
000183
000184
000185  ;-----------------------------------------------------------------------
000186  ;
000187  ;  Zero Page Storage
000188  ;
000189  ;-----------------------------------------------------------------------
000190
000191  ZPGSAVE         .EQU         SOSINT+0A                 ;Saved zero page storage
000192
000193  ZPGTEMP         .EQU         ZPGSAVE+00                ;Temporary zero page storage
000194  MOVCNT          .EQU         ZPGTEMP+00
000195
000196
000197  ;-----------------------------------------------------------------------
000198  ;
000199  ;   Private Variable Storage
000200  ;
000201  ;-----------------------------------------------------------------------
000202
000203  SIRADDR         .WORD        SIRTABLE
000204  SIRTABLE        .BYTE        1,0                       ;ACIA resource
000205                  .WORD        ACIAMIH
000206  MIHBANK         .BYTE        0
000207  SIRCOUNT        .EQU         *-SIRTABLE
000208  OPENFLG         .BYTE        FALSE                     ;Device open flag
000209  XMIT            .BYTE        FALSE                     ;XMIT in progress flag
000210  DLYCNT          .BYTE        0                         ;Delay count for MIH
000211  BUFCNT          .BYTE        0                         ;Local buffer byte count
000212  BUFHEAD         .BYTE        0                         ;Local buffer head index
000213  BUFTAIL         .BYTE        0                         ;Local buffer tail index
000214  BUFSIZE         .EQU         110.                      ;Local buffer size
000215  LOCBUF          .EQU         *                         ;Local buffer
000216                  .ASCII       "Copyright (C) 1983 by Apple Computer Inc."
000217  CPYRGHTSIZ      .EQU         *-LOCBUF
000218                  .BLOCK       BUFSIZE-CPYRGHTSIZ,0
000219                  .PAGE
000220  ;-----------------------------------------------------------------------
000221  ;
000222  ;  Serial Printer Driver -- Main entry point
000223  ;
000224  ;-----------------------------------------------------------------------
000225
000226  SP_MAIN         .EQU         *
000227                  SWITCH       REQCODE,8,SP_REQSW
000228
000229
000230  BADREQ          LDA          #XREQCODE                 ;Invalid request code
000231                  JSR          SYSERR
000232
000233
000234  NOTOPEN         LDA          #XNOTOPEN                 ;Device not open
000235                  JSR          SYSERR
000236
000237
000238  SP_REQSW        .EQU         *                         ;Serial Printer request switch
000239                  .WORD        SP_READ-1
000240                  .WORD        SP_WRITE-1
000241                  .WORD        SP_STAT-1
000242                  .WORD        SP_CNTL-1
000243                  .WORD        BADREQ-1
000244                  .WORD        BADREQ-1
000245                  .WORD        SP_OPEN-1
000246                  .WORD        SP_CLOSE-1
000247                  .WORD        SP_INIT-1
000248                  .PAGE
000249  ;----------------------------------------------------------------------
000250  ;
000251  ;  Serial Printer Driver -- Initialization Request
000252  ;
000253  ;----------------------------------------------------------------------
000254
000255  SP_INIT         .EQU         *
000256                  LDA          #FALSE
000257                  STA          OPENFLG
000258                  LDA          DRATE                     ;Validate data rate
000259                  AND          #00F
000260                  STA          DRATE
000261                  TAX
000262                  LDA          DFORMAT                   ;Validate data format
000263                  AND          #0EE
000264                  ORA          #010
000265                  CPX          #03                       ;If data rate is 110 baud
000266                  BNE          $010
000267                  ORA          #080                      ;  force two stop bits
000268  $010            STA          DFORMAT
000269                  CLC
000270                  RTS

000271                  .PAGE
000272  ;----------------------------------------------------------------------
000273  ;
000274  ;  Serial Printer Driver -- Open Request
000275  ;
000276  ;----------------------------------------------------------------------
000277
000278  SP_OPEN         .EQU         *
000279                  BIT          OPENFLG                   ;Serial Printer open?
000280                  BPL          $010                      ;  No
000281                  LDA          #XNOTAVIL
000282                  JSR          SYSERR
000283
000284  $010            LDA          B_REG
000285                  AND          #0F
000286                  STA          MIHBANK                   ;Set interrupt handler bank
000287                  LDA          #SIRCOUNT
000288                  LDX          SIRADDR
000289                  LDY          SIRADDR+1
000290                  JSR          ALLOCSIR                  ;Allocate the ACIA
000291                  BCS          $020
000292
000293                  LDA          #FALSE
000294                  STA          XMIT
000295                  JSR          CNTL00                    ;Set up ACIA
000296                  LDA          #TRUE
000297                  STA          OPENFLG                   ;Set serial printer open
000298                  RTS
000299
000300  $020            LDA          #XNORESRC
000301                  JSR          SYSERR
000302                  .PAGE
000303  ;----------------------------------------------------------------------
000304  ;
000305  ;  Serial Printer Driver -- Close Request
000306  ;
000307  ;----------------------------------------------------------------------
000308
000309  SP_CLOSE        .EQU         *
000310                  ASL          OPENFLG                   ;Serial Printer open?
000311                  BCS          $010                      ;  Yes
000312                  JMP          NOTOPEN
000313
000314  $010            BIT          XMIT                      ;Wait for write completion
000315                  BMI          $010
000316                  PHP
000317                  SEI
000318                  LDA          E_REG
000319                  TAX
000320                  ORA          #BITON7
000321                  STA          E_REG                     ;Switch to 1 MHz
000322                  STA          ACIASTAT                  ;Reset the ACIA
000323                  STX          E_REG
000324                  PLP
000325                  LDA          #SIRCOUNT
000326                  LDX          SIRADDR
000327                  LDY          SIRADDR+1
000328                  JSR          DEALCSIR                  ;Deallocate the ACIA
000329                  RTS
000330                  .PAGE
000331  ;-----------------------------------------------------------------------
000332  ;
000333  ;  Serial Printer Driver -- Read Request
000334  ;
000335  ;-----------------------------------------------------------------------
000336
000337  SP_READ         .EQU         *
000338                  BIT          OPENFLG                   ;Serial Printer open?
000339                  BMI          $010
000340                  JMP          NOTOPEN
000341  $010            LDA          #XBADOP
000342                  JSR          SYSERR
000343                  .PAGE
000344  ;-----------------------------------------------------------------------
000345  ;
000346  ;  Serial Printer Driver -- Write Request
000347  ;
000348  ;-----------------------------------------------------------------------
000349
000350  SP_WRITE        .EQU         *
000351                  BIT          OPENFLG
000352                  BMI          $010
000353                  JMP          NOTOPEN
000354  $010            LDA          #BUFSIZE/2                ;Set MOVCNT to the lesser
000355                  LDY          REQCNT+1                  ;  of BUFSIZE/2 and REQCNT.
000356                  BNE          $020
000357                  CMP          REQCNT
000358                  BCC          $020
000359                  LDA          REQCNT
000360                  BNE          $020
000361                  RTS                                    ;Count = zero -- all done!
000362  $020            STA          MOVCNT
000363

 Apple /// Serial Printer Port Driver 1.30 Source Code Listing  ---  5 / 8



000364                  LDA          BUFFER+1                  ;Check for buffer
000365                  CMP          #0FF                      ;  address overflow
000366                  BCC          $030
000367                  SBC          #080
000368                  STA          BUFFER+1
000369                  INC          1401+BUFFER
000370
000371  $030            SEC
000372                  LDA          #BUFSIZE
000373                  SBC          MOVCNT
000374  $040            CMP          BUFCNT                    ;Wait for room in buffer
000375                  BCC          $040
000376
000377                  LDY          #0
000378                  LDX          BUFTAIL
000379  $050            LDA          (BUFFER),Y                ;Move data to local buffer
000380                  STA          LOCBUF,X
000381                  INX
000382                  CPX          #BUFSIZE
000383                  BCC          $060
000384                  LDX          #0
000385  $060            INY
000386                  CPY          MOVCNT
000387                  BCC          $050
000388                  STX          BUFTAIL
000389
000390                  PHP
000391                  SEI                                    ;Shut down interrupts
000392                  CLC
000393                  LDA          BUFCNT
000394                  ADC          MOVCNT                    ;Bump buffer count
000395                  STA          BUFCNT
000396
000397                  BIT          XMIT                      ;Already transmitting?
000398                  BVS          $070                      ;  Yes
000399                  LDA          #0C0
000400                  STA          XMIT                      ;Set transmitting flag
000401                  LDA          E_REG
000402                  PHA
000403                  ORA          #BITON7                   ;Switch to 1 MHz
000404                  STA          E_REG
000405                  LDY          ACIASTAT                  ;Fake an interrupt to start
000406                  JSR          ACIAMIH                   ;  the interrupt handler.
000407                  PLA
000408                  STA          E_REG                     ;Switch back to 2 MHz
000409  $070            PLP
000410
000411                  CLC
000412                  LDA          BUFFER
000413                  ADC          MOVCNT                    ;Fix up buffer pointer
000414                  STA          BUFFER
000415                  BCC          $080
000416                  INC          BUFFER+1
000417
000418  $080            SEC
000419                  LDA          REQCNT
000420                  SBC          MOVCNT                    ;Fix up requested count
000421                  STA          REQCNT
000422                  BCS          $010
000423                  DEC          REQCNT+1
000424                  JMP          $010                      ;Loop back for more
000425                  .PAGE
000426  ;-----------------------------------------------------------------------
000427  ;
000428  ;  ACIA Master Interrupt Handler
000429  ;
000430  ;-----------------------------------------------------------------------
000431
000432  ACIAMIH         .EQU         *
000433                  LDA          E_REG
000434                  ORA          #BITON7                   ;Set 1 MHz mode
000435                  STA          E_REG
000436
000437                  TYA                                    ;Check DSR and DCD status
000438                  AND          #60                       ;  bits for printer hand shake
000439                  BNE          $080
000440
000441                  TYA                                    ;Check transmit register
000442                  AND          #BITON4                   ;  empty status bit
000443                  BEQ          $060
000444
000445                  LDA          DLYCNT                    ;Any transmit delay in progress?
000446                  BEQ          $010                      ;  no
000447                  DEC          DLYCNT
000448                  JMP          $060
000449
000450  $010            LDA          BUFCNT                    ;Any data to transmit?
000451                  BEQ          $070                      ;  no -- wait for completion
000452                  LDX          BUFHEAD
000453                  LDA          LOCBUF,X
000454                  STA          ACIADATA                  ;Transmit one character
000455                  INX
000456                  CPX          #BUFSIZE

000457                  BCC          $020
000458                  LDX          #0
000459  $020            STX          BUFHEAD                   ;Update buffer index
000460                  DEC          BUFCNT                    ;  and count
000461
000462                  CMP          #ASC_CR                   ;Check for any delay
000463                  BEQ          $040
000464                  BCS          $060
000465                  CMP          #ASC_LF
000466                  BNE          $030
000467                  LDA          LFDELAY
000468                  BCS          $050
000469  $030            CMP          #ASC_FF
000470                  BNE          $060
000471                  LDA          FFDELAY
000472                  BCS          $050
000473  $040            LDA          CRDELAY
000474  $050            STA          DLYCNT
000475                  .PAGE
000476  $060            LDA          ACIACMD
000477                  AND          #0E0                      ;Enable transmit interrupt
000478                  ORA          #007
000479                  STA          ACIACMD
000480                  RTS
000481
000482  $070            ASL          XMIT
000483                  BMI          $060                      ;Still not done
000484
000485  $080            LDA          ACIACMD
000486                  AND          #0E0                      ;Disable transmit interrupt
000487                  ORA          #00B
000488                  STA          ACIACMD
000489                  RTS
000490                  .PAGE
000491  ;----------------------------------------------------------------------
000492  ;
000493  ;  Serial Printer Driver -- Status Request
000494  ;
000495  ;----------------------------------------------------------------------
000496
000497  SP_STAT         .EQU         *
000498                  BIT          OPENFLG                   ;Serial Printer open?
000499                  BMI          $010
000500                  JMP          NOTOPEN
000501  $010            SWITCH       CTLSTAT,2,STATSW
000502
000503
000504  BADCTL          LDA          #XCTLCODE                 ;Invalid control code
000505                  JSR          SYSERR
000506
000507
000508  STATSW          .WORD        STAT00-1
000509                  .WORD        STAT01-1
000510                  .WORD        STAT02-1
000511
000512
000513  STAT00          RTS                                    ;0 -- NOP
000514
000515
000516  STAT01          LDY          #0                        ;1 -- Status Table
000517                  LDA          #0
000518                  STA          (CSLIST),Y
000519                  RTS
000520
000521
000522  STAT02          LDY          #0                        ;2 -- New Line
000523                  LDA          #FALSE
000524                  STA          (CSLIST),Y
000525                  RTS
000526                  .PAGE
000527  ;----------------------------------------------------------------------
000528  ;
000529  ;  Serial Printer Driver -- Control Request
000530  ;
000531  ;----------------------------------------------------------------------
000532
000533  SP_CNTL         .EQU         *
000534                  BIT          OPENFLG                   ;Serial Printer open?
000535                  BMI          $010                      ;  Ok
000536                  JMP          NOTOPEN
000537  $010            SWITCH       CTLSTAT,2,CNTLSW
000538                  JMP          BADCTL
000539
000540  CNTLSW          .WORD        CNTL00-1
000541                  .WORD        CNTL01-1
000542                  .WORD        CNTL02-1
000543
000544  CNTL00          .EQU         *                         ;0 -- Reset
000545  $010            BIT          XMIT                      ;Wait for write completion
000546                  BMI          $010
000547                  LDA          #00
000548                  STA          BUFHEAD
000549                  STA          BUFTAIL

000550                  PHP
000551                  SEI
000552                  LDA          E_REG
000553                  TAX
000554                  ORA          #BITON7
000555                  STA          E_REG                     ;Switch to 1 MHz
000556                  STA          ACIASTAT                  ;Reset ACIA
000557                  LDA          DFORMAT
000558                  AND          #0F0
000559                  ORA          DRATE
000560                  STA          ACIACTL                   ;Set up ACIA control register
000561                  LDA          DFORMAT
000562                  ASL          A
000563                  ASL          A
000564                  ASL          A
000565                  ASL          A
000566                  ORA          #00B
000567                  STA          ACIACMD                   ;Set up ACIA command register
000568                  STX          E_REG                     ;Switch back to 2 MHz
000569                  PLP
000570                  RTS
000571
000572  CNTL01          .EQU         *                         ;1 -- Serial Printer Status Table
000573                  RTS
000574
000575  CNTL02          .EQU         *                         ;2 -- New Line
000576                  RTS
000577                  .END
000578

; ####################################################################################################
; #   END OF FILE:  SERPRINT.text
; #   LINES      :  578
; #   CHARACTERS :  26582
; #   Formatter  :  Assembly Language Reformatter 1.0.2 (07 January 1998)
; #   Author     :  David T. Craig -- 71533.606@compuserve.com -- Santa Fe, New Mexico USA
; ####################################################################################################

###

